A friendly Fortran DDE solver

S. Thompson a,*, L.F. Shampine b

a Department of Mathematics & Statistics, Radford University, Radford, VA 24142, USA
b Mathematics Department, Southern Methodist University, Dallas, TX 75275, USA

Available online 24 May 2005

Abstract

DKLAG6 is a FORTRAN 77 code widely used to solve delay differential equations (DDEs). Like all the popular Fortran DDE solvers, new users find it formidable and in many respects, it is not easy to use. We have applied our experience writing DDE solvers in MATLAB and the capabilities of Fortran 90 to the development of a friendly Fortran DDE solver. How we accomplished this may be of interest to others who would like to modernize legacy code. In the course of developing a completely new user interface, we have added significantly to the capabilities of DKLAG6.

© 2005 IMACS. Published by Elsevier B.V. All rights reserved.

Keywords: Delay differential equations; State dependent delays; Discontinuity tracking; Mathematical software

1. Introduction

This investigation is another step in a line of software development that began with the first ordinary differential equation (ODE) solver, $\text{ode45}$, of the problem solving environment (PSE) MATLAB [7]. $\text{ode45}$ is a translation of the FORTRAN solver RKF45 [14], but it is not merely a translation: C.B. Moler defined a new user interface to make solving ODEs as easy as possible. In this he exploited the capabilities of the MATLAB language and recognized the kinds of problems typical of the PSE. Later Shampine and Reichelt [11] extended this interface to increase the capabilities of the software and accommodate other methods for solving ODEs. In particular, all the solvers of the ODE Suite were endowed with a powerful event location capability. Programs for solving delay differential equations (DDEs) have been
much harder to use than programs for solving ODEs, in part because the task is itself more complicated. We firmly believe, however, that more scientists would use DDE models if it were easier to solve them numerically. This led us to develop the solver \texttt{dde23} [13] that is now part of MATLAB. It is restricted to problems with constant delays, but given that, it makes solving simple problems easy and it solves conveniently problems with complications like event location and restarts. The \texttt{ddesd} program [9] solves DDEs with time- and state-dependent delays. Though the underlying method is quite different from that of \texttt{dde23}, the user interface is as similar as the tasks allow. For brevity we refer to these MATLAB DDE solvers as \texttt{dde23/sd}.

The DDE solver DKLAG6 [3] has capabilities that are comparable to those of \texttt{dde23/sd} and solves a larger class of problems, but new users find it formidable and in many respects, it is not easy to use. We believe that it is fair to say this about all the widely-used, general-purpose DDE solvers written in FORTRAN. Our principal goal was to develop a program, DDE_SOLVER, based on DKLAG6 that approaches the convenience and ease of use of \texttt{dde23/sd}. DKLAG6 is written in FORTRAN 77, but to reach our goal, we had to have the extra features of Fortran 90/95. For brevity we write F77 to indicate FORTRAN 77 and F90 to indicate Fortran 90/95. A friendly user interface is already enough to justify our effort, but we have also made important qualitative improvements to DKLAG6.

We are mainly interested in solving retarded DDEs for which the (vector) solution \(y(t)\) satisfies

\[
y'(t) = f(t, y(t), y(\beta_1), \ldots, y(\beta_k))
\]

for \(t_0 \leq t \leq t_f\). The terms \(y(\beta_j) = y(\beta_j(t, y(t)))\) are the solution evaluated at delayed arguments. They are delayed because we require that the delay functions \(\beta_j\) all satisfy \(\beta_j(t, y(t)) \leq t\) for the solution vector \(y(t)\). A solution must also satisfy

\[
y(t) = h(t)
\]

for \(t < t_0\) and a given history function \(h(t)\). Although DDE_SOLVER allows time- and state-dependent delay functions, the most common form involves a constant lag \(\tau_j > 0\), namely

\[
\beta_j(t, y(t)) = t - \tau_j
\]

DDE_SOLVER makes special provision for problems with constant lags because they are very common, they are considerably easier to solve, and the user interface can be simplified.

Along with the integration of the DDEs, we may be interested in locating where any one of a collection of event functions,

\[
g_j(t, y(t), y'(t))
\]

vanishes. The places where they vanish are called events. Sometimes we just want the solver to report the location of an event and the solution there. Other times we want to terminate the integration or restart the integration after changing the problem. Not very many DDE solvers provide for event location. DKLAG6 does, and we have augmented its capabilities for DDE_SOLVER.

In Section 8 we illustrate the solution of an equation of retarded type used to model lasers. The one delay function involves a constant lag and the history is constant. There is an event function. DDE_SOLVER is much faster than \texttt{dde23} and not much harder to use. DDE_SOLVER, a collection of example programs, and an expanded version of this paper are available at the web site [5]. The typical program writes the solution to a file. In most cases the F90 program is accompanied by an M-file that imports the numerical solution and plots it in MATLAB.
2. Simple calls for simple problems

One of our goals was to make solving simple problems as easy as possible. As it has turned out, all you have to do is define the problem: A problem of the form (1), (2), (3) is solved with a call like

\[ \text{SOL} = \text{DDE_SOLVER}(\text{NVAR}, \text{DDES}, \text{BETA}, \text{HISTORY}, \text{TSPAN}) \]  

Here NVAR is an integer array of two entries. The first is NEQN, the number of equations in (1), and the second is NLAGS, the number of delays in (3). DDES is the name of a subroutine for evaluating (1). It has the form

\[ \text{SUBROUTINE DDES}(T, Y, Z, DY) \]

The input arguments are the independent variable T, a vector Y of NEQN components approximating \( y(T) \), and an array Z that is NEQN \( \times \) NLAGS. Column \( j \) of this array is an approximation to \( y(\beta_j(T, y(T))) \). The subroutine is to evaluate (1) with these arguments and return \( y'(T) \) as the vector DY of NEQN components.

BETA is the name of a subroutine for evaluating the delays and HISTORY is the name of a subroutine for evaluating (2). To be more precise, the functions are defined by subroutines for general problems, but they are defined in a simpler way in the very common situations of constant lags and/or constant history. How this is done is explained in Section 4.

Output is handled very differently in DKLAG6 and DDE_SOLVER. Both codes have a number of possibilities, but here we describe only the default modes. DDE_SOLVER returns the numerical solution in the output structure SOL. The input vector TSPAN is used to inform the solver of the interval of integration and where approximate solutions are desired. TSPAN has at least two entries. The first entry is the initial point of the integration, \( t_0 \), and the last is the final point, \( t_f \). If TSPAN has only two entries, approximate solutions are returned at all the mesh points selected by the solver itself. These points generally produce a smooth graph when the numerical solution is plotted. If TSPAN has entries \( t_0 < t_1 < \cdots < t_f \), the solver returns approximate solutions at (only) these points.

The call list of (5) resembles closely that of \texttt{dde23/sd}. Our design provides for a considerable variety of additional capabilities. This is accomplished in two ways. F90 provides for optional arguments that can be supplied in any order if associated with a keyword. This is used, for example, to pass to the solver the name of a subroutine EF for evaluating (4) with a call like

\[ \text{SOL} = \text{DDE_SOLVER}(\text{NVAR}, \text{DDES}, \text{BETA}, \text{HISTORY}, \text{TSPAN}, \text{EVENT_FCN=}EF) \]

Details are provided below. One of the optional arguments is a structure containing options. This structure is formed by a function called DDE_SET that is analogous to the function \texttt{ddeset} used by \texttt{dde23/sd}. The call list of (5) uses defaults for important quantities such as error tolerances, but of course, the user has the option of specifying quantities appropriate to the problem at hand. These options are discussed in sections that follow.

3. New answers to old questions

In this section we explain how a number of the software issues in DKLAG6 can be handled in a more satisfactory way by exploiting F90. Other convenient aspects of F90 are illustrated by the numerical
example of Section 8 where quantities are initialized where they are declared and arrays are formed using array constructors.

3.1. Precision

MATLAB does all computations in a working precision. This contrasts with Fortran for which it is necessary to select a precision for REAL variables. DKLAG6 is coded for double precision. At the time DKLAG6 was written, double precision arithmetic could have a quite different meaning on computers in wide use. A generally accepted way to deal with this was to call a subroutine D1MACH to obtain the hardware characteristics, and that is what is done in DKLAG6. F90 handles this in a much more satisfactory way by providing intrinsics for the purpose in the language itself. For instance, the unit roundoff corresponding to double precision is obtained by UROUND = EPSILON(1D0). F90 makes it possible to code software so that the precision can be changed easily, but this complicates the way that users have to write their subroutines. Also, the IEEE arithmetic standard has become generally accepted and with it, double precision is by far the usual choice for scientific computing. For these reasons DDE_SOLVER is coded entirely in double precision and it assumes that the user will provide subroutines coded in a similar fashion.

3.2. Array operations

DKLAG6 makes use of several of the BLAS [6]. These F77 subroutines were developed to clarify vector operations by encapsulating them and to speed up the operations by means of implementations that exploit the hardware. Their functionality is now directly available in F90. For instance, a vector K2 of length N is multiplied by a scalar ALPHA and added to a vector R of the same length by CALL DAXPY(N,ALPHA,K2,1,R,1). With the array operations of F90 this can be expressed in a more straightforward way as R = ALPHA*K2 + R. This feature allowed us to replace numerous loops throughout the code with much simpler statements.

3.3. Storage

A basic difficulty in writing mathematical software in F77 is to manage storage in a way convenient for users. We address the issue of not knowing in advance how much storage will be needed in the next section and here take up only provision of a fixed amount of storage. DKLAG6 requires a good many vectors of working storage that have a length known in advance, generally NEQN or NLAGS. These vectors are of no interest to the user, so making them arguments in the call list of DKLAG6 would make the call list intolerably long. The standard way of dealing with this difficulty was to have the user supply a work array with a length given by a formula involving NEQN (and possibly other parameters). Segments of this array are used for the various vectors of the algorithm. If this were done directly, the algorithm would be unreadable, so pointers are defined on entry to DKLAG6 and segments of the work array are used as arguments in calls to lower level subroutines. The lower level subroutines have all the requisite vectors in their call lists. In this way the vectors of the algorithm all have meaningful names, but the user is exposed only to the work array at the top level. This can all be handled more conveniently in F90 by means of global storage. In the specification part of the module for the solver we declare the various vectors as allocatable arrays. This makes them visible to all the subroutines of the module. By
giving them the PRIVATE attribute, they are not visible outside the module. Using global arrays we have simplified the program notably by doing away with the passing of a great many vectors through several levels of subroutines.

3.4. Dynamic storage

An obvious difference between DKLAG6 and DDE_SOLVER is that the user does not have to consider storage at all. This is much more than a mere convenience; it is a qualitative improvement in the software. That is because it is not known in advance how much storage will be needed for some of the arrays. One example is the queue that holds the discontinuity tree—the number of discontinuities depends on the problem. A person using DKLAG6 must guess how much storage will be required and if the guess is not big enough, the computation will fail.

F90 has dynamic storage capabilities and array operations that we use in DDE_SOLVER to deal with the work arrays and queue. In the first instance this is a matter of making these arrays ALLOCATABLE and an initial allocation of enough storage to solve an easy problem. Of course it might turn out that more storage will be needed to complete the computation. The difficulty then is that we need to increase the sizes of arrays whilst retaining the values stored in the arrays. For reasons of efficiency, we increase the sizes of the arrays by a substantial amount. If we could return an allocatable array, we would proceed similarly with the solution itself, but that is not allowed in F90. Instead we define a derived type DDE_SOL with fields like

\[
\text{DOUBLE PRECISION, DIMENSION(:,:), POINTER :: Y}
\]

and return a structure SOL of this type. In the solver we can allocate storage for the field SOL\%Y and store the solution in it as the integration proceeds. Like the working storage we increase the storage as necessary in chunks of substantial size, but before returning we trim the array to the size actually needed for the solution.

4. Constant functions

In general the initial history is defined by a function, but a constant history is common in practice. For this reason if all components of the history are constant, DDE_SOLVER allows users to supply a constant history in the form of a vector rather than making them write a subroutine that returns a constant vector. Similarly, a user must generally supply a function that evaluates the delayed arguments \( \beta_j(t, y(t)) \). However, it is common in practice that all the delayed arguments have the form \( t - \tau_j \) for constant lags \( \tau_j \). For such problems it is convenient to supply the vector of lags and have the solver form the delayed arguments as needed. Constant lags greatly simplifies construction of the discontinuity tree and in other ways leads to a more efficient and accurate integration.

Simpler calls in common circumstances are achieved in F90 by writing several versions of the solver that users call with the same (generic) name, DDE_SOLVER. In F90 the number and type of input arguments in a given call are compared to the various versions and the matching version of the solver is selected for execution. All our versions of DDE_SOLVER call the same underlying code, a modification of DKLAG6. If, say, HISTORY is supplied as a vector, the version of the solver that expects a vector
argument is selected. In this way we make the solution of DDEs more efficient and easier to use in the common circumstances of constant history and/or constant lags.

5. Optional arguments

F90 allows optional arguments to functions and subroutines that follow the required arguments in the call list. Though optional arguments can be passed by position in the list, we assume that they will always be identified with keywords, another new possibility in F90. Using keywords, we can set just the options of interest and we can set them in any order. This was illustrated in Section 2 with the EVENT_FCN option. The options that are handled in this way are discussed below. We begin with specification of a variety of options using an options structure and the optional argument OPTIONS.

5.1. Options

To simplify the solution of DDEs, many of the options allowed by dde23/sd are grouped in an options structure. The structure is an optional argument to the solver and if it is not present, defaults are used for all the options. The options structure is formed by an auxiliary function, ddeset. In an invocation of this function, the user sets any options he wishes and any options not set are given default values. It is possible to do something quite similar in F90. We have defined a derived type DDE_OPTS with fields for the various options. The auxiliary function DDE_SET is used to set the options. The PRESENT intrinsic is used inside DDE_SET to determine whether the user has set an optional argument and if not, a default value is given to the field. An options structure is an optional argument in DDE_SOLVER. All arguments of DDE_SET are optional, so if the user does not provide an options structure to the solver, it will use OPTIONS = DDE_SET() to form an options structure with all fields set to default values.

We discuss more fully the error tolerances because they are the options most commonly set by the user and they present some points of interest. DKLAG6 has relative and absolute error tolerances that are vectors. Tolerances must be specified for DKLAG6, but they are optional arguments in the call to DDE_SET. If the scalar relative error tolerance RE is not specified, it is given a default value of $10^{-3}$. Absolute error tolerances are more complicated. The most common case is the default, which is to set all NEQN components of the absolute error tolerance vector to $10^{-6}$. It is a temptation to forego the remaining case of a vector tolerance, but this is not possible because a vector tolerance is actually needed for a good many problems. To facilitate the second case, DKLAG6 allows users to specify a vector of only one component. However, to do this the user must also give an input argument ITOL a value that instructs DKLAG6 to expand this vector to one of length NEQN. The inconvenience of setting ITOL can be eliminated in F90 by using the SIZE intrinsic to recognize the case, but declaring a vector tolerance of length one is unnatural when solving a system of NEQN equations. This is handled better in dde23/sd because MATLAB does not distinguish a scalar and a vector of one component. We have not been able to do the same in F90, so we have resorted to two options for absolute error tolerances in DDE_SET. Corresponding to the scalar RE option there is a scalar AE option. In addition there is a vector option called AE_VECTOR. If no absolute error tolerance is specified, the default is used. If the scalar option AE is set, the value input is assigned to all NEQN components of an absolute error tolerance vector used by the solver. If the vector option AE_VECTOR is set, the vector (of NEQN components) that is input is used for the absolute error tolerance vector.
5.2. Event location

In our experience event location has been crucial to the effective solution of many problems. It is not widely available, certainly not in DDE solvers, because of practical and theoretical difficulties discussed in [10,12]. We considered event location to be an essential capability for DKLAG6 [3]. It was a explicit design goal for the MATLAB ODE Suite [11], so there was never any question that dde23/sd would have it. The MATLAB solvers have a more powerful capability for event location than DKLAG6, so we provided their additional features in DDE_SOLVER. In most respects the interface for event location in DDE_SOLVER resembles closely that of dde23/sd. In particular, both solvers return the same information about events, viz., the location, the numerical solution at the event, and an integer identifying which event occurred, as fields in the solution structure. However, the event functions (4) allowed by DDE_SOLVER are more general than those of dde23/sd in that they can depend on $y'(t)$. And, some matters are handled in a different way.

Event location is optional, so the solver must be told that it is to locate zeros of a collection of functions evaluated in a given procedure. In dde23/sd the user does this by providing a function handle as the value of the option Events using ddeset. In DDE_SOLVER the user provides the name of the subroutine as an optional input argument of the solver itself that is specified with the keyword EVENT_FCN.

It can be useful to distinguish how an event function crosses the axis. For example, this can be used to deal with problems that have events at the initial point. In the example of Section 8 it is used to distinguish a maximum from a minimum when the first derivative of a solution component is an event function. In dde23/sd the function for evaluating events must return a vector that states what the user wants. Component $j$ of a vector direction is given the value $-1$ when event $j$ is interesting only if the event function decreases through 0, the value $+1$ when it is interesting only if the function increases through 0, and the value 0 when it does not matter how the function crosses the axis. We added this capability to DDE_SOLVER, but the vector DIRECTION is set as an option using DDE_SET. It has a default value with all components equal to 0. This is different from dde23/sd, which does not have a default for direction. Adding this feature to the new solver was not merely a matter of adding it to the interface; DKLAG6 does not have this feature, so the algorithm for locating events had to be augmented appropriately.

A very important distinction is made between terminal and non-terminal events. As the name suggests, a terminal event causes the program to finish up and return control to the user. In dde23/sd the function for evaluating events must return a vector isterminal. A value of 1 in component $j$ means that the integration is to terminate if the event function (4) crosses the axis in the specified direction, and a value of 0 means that information about the event is to be made available, but the integration is to continue. Coding in F90 it was more natural to make the vector ISTERMINAL of type LOGICAL.

DKLAG6 and dde23/sd take fundamentally different approaches to terminal events. To understand the issues, a concrete example will be helpful. In [13] we discuss the solution of DDEs modelling the behavior of a two-wheeled suitcase [15] with DKLAG6 and dde23. As the suitcase rolls along, it can start to rock from one wheel to the other. When a wheel hits the ground, the integration must be restarted with new initial conditions that reflect the wheel bouncing, i.e., with a change of direction and a speed reduced by a coefficient of restitution. In MATLAB it was convenient to store as fields in the solution structure returned by dde23/sd all the information needed to restart from the last point reached. Coding this is complicated because of all the possibilities, but from a user’s point of view, restarting is merely a matter of supplying a solution structure as history rather than a function (or vector). We can solve the
suitcase problem in this design by making a wheel hitting the ground a terminal event. If the solver returns because a wheel hit the ground, we restart the integration with initial conditions determined appropriately from the solution returned at the time of the event.

DKLAG6 does not provide for restarts and after careful consideration, we decided not to add this capability to DDE_SOLVER. One reason is that it would be a major change to the legacy code. Another is that it is not clear to us how to code something using F90 that resembles closely what we did in dde23/sd. Finally, the approach taken in DKLAG6 is quite satisfactory; it is just that the user interface is unduly complicated. In our new design the user writes a subroutine of the form

\[
\text{CHANGE(NE, } T, Y, DY, \text{HINIT, DIRECTION, ISTERMINAL, QUIT)}
\]

and informs the solver of this subroutine by means of an optional input argument with keyword CHANGE_FCN. The solver calls this subroutine at every event. In this call NE is an integer that identifies which event occurred, T is the location of the event, and Y and DY are the approximations to the solution and its first derivative, respectively, at the event. In CHANGE the user can inspect this information and decide what, if any, action is appropriate. For instance, he might want to redefine the DDEs. It is good practice to define the DDEs and auxiliary subroutines such as CHANGE and EVENTS in a module. This makes possible variables that are available to all the subroutines of the module. It is then easy to use such a variable in the DDEs subroutine to determine how the equations are to be evaluated and to change the value of this variable in CHANGE in response to an event. Y and DY are also output variables that the user can reset. That is just what is needed to solve the suitcase problem. In effect we restart the integration with new initial values. The output variable HINIT allows a user to specify a step size for the solver to try on return. The integer array DIRECTION and the logical array ISTERMINAL can be changed so that the event functions will be interpreted differently on return from CHANGE. The integration can be terminated by changing QUIT from its input value of .FALSE. to .TRUE.

6. Form of the solution

As already mentioned in Section 2, the default is to return the approximate solution at the mesh points selected by DDE_SOLVER. If the solution structure is called SOL, the number of mesh points NPTS is returned as SOL%NPTS, the mesh points \( t_0 < t_1 < \cdots < t_f \) chosen by the solver are returned as SOL%T, and the corresponding approximate solutions are returned in the NPTS \( \times \) NEQN array SOL%Y. The only difference when TSPAN has more than two entries is that SOL%T is then the same as TSPAN.

There are two other ways that the solver can return a solution. A general-purpose DDE solver must have a continuous extension that allows the solution to be approximated accurately between mesh points. The default output of dde23/sd is to return in the solution structure the information needed to evaluate these continuous extensions anywhere in the interval of integration. We have endowed DDE_SOLVER with this capability. Because the capability requires a considerable amount of additional information in SOL, we have made this an option called INTERPOLATION. By default it has the value .FALSE. If DDE_SET is used to set this option to .TRUE., the solver includes in SOL the information needed for interpolation. A call to the auxiliary function DDE_VAL of the form

\[
\text{YINT = DDE VAL(T, SOL)}
\]
evaluates approximations to \( y(T) \) for any \textit{vector} of points \( T \) that all lie in \([t_0, t_f]\). The values \( y'(T) \) are also returned when an optional argument of DDE_VAL with keyword \textsc{derivatives} is set to .TRUE. Sometimes only selected components are of interest. The optional argument with keyword \textsc{components} is a vector that tells DDE_VAL which components of the solution are desired. For example, the first and third components of the solution at the two arguments \( t = 3.14, 1.41 \) would be returned by

\[
\text{YINT} = \text{DDE VAL}((/ 3.14D0, 1.41D0 /), \text{SOL}, \text{COMPONENTS}=(/ 1, 3 /))
\]

The output argument is a derived type called \textsc{dde int}, so \textsc{YINT} in this example must be declared as

\[
\text{TYPE} (\text{DDE INT}) :: \text{YINT}
\]

Although the vector \( T \) is available in the calling program, it is convenient to return it as the field \textsc{YINT%TVALS} in the interpolation structure. Similarly, if a vector is provided for the keyword \textsc{components}, it is returned as the field \textsc{YINT%COMPONENTS}. If this option is not set, the field is given the default value of all the components, namely the vector with entries 1, \ldots, \textsc{neqn}. The approximate solution is returned in the field \textsc{YINT%YT}. More specifically, \textsc{YINT%YT(I,J)} is an approximation to component \( J \) of the solution at \( T(I) \). Similarly, if \textsc{derivatives}=.TRUE., an approximation to component \( J \) of the first derivative of the solution at \( T(I) \) is returned in \textsc{YINT%DT(I,J)}.

7. Discontinuity tracking

A fundamental issue is whether to track discontinuities explicitly or to rely on the error control to handle them indirectly. In the latter approach it is hoped that the error estimator will recognize discontinuities and cause the solver to reduce the step size automatically to get past them without degrading the accuracy of the solution too much. The \texttt{ddesd} solver [9] takes a different approach. It does not track discontinuities, but it is intended only for problems with solutions that get smoother as the integration proceeds, and it estimates and controls the residual in an attempt to handle discontinuities more robustly than conventional local error control. Other solvers that attempt to deal directly with discontinuities do so in different ways. Since it is limited by design to DDEs with constant lags, \texttt{dde23} [13] builds the discontinuity tree at the beginning of the integration and then steps exactly to each point in the tree (up to the order of the Runge–Kutta method used in the code) to avoid interpolating across a discontinuity. The situation is more troublesome when the delays are time- and state-dependent. Although none of the tracking methods in use is entirely satisfactory, they all work well enough in practice and significantly improve both the efficiency and the reliability of the solvers. \texttt{DDVERK} [4] uses a method based on the behavior of the solver following error test failures. When a discontinuity is suspected, bisection is used to locate it and then special interpolants are used to get past it. The user of \texttt{ARCHI} [8] can provide a diagram that defines the dependence of the DDEs on the various delays. The solver uses it to track the discontinuities explicitly, locating them precisely with a root finding subroutine. When all delays are constant, \texttt{DDE_SOLVER} builds the tree in a manner similar to \texttt{dde23} and steps to each point in the tree exactly, thereby eliminating the need for root finding as well as the necessity to use questionable error estimates near discontinuities. When the delays are state-dependent, \texttt{DDE_SOLVER} tracks discontinuities explicitly using root finding in the manner described below.
Because we believe that tracking discontinuities should be invisible to the user, DDE_SOLVER does not ask for any information about discontinuities, with one minor exception. We assume that there is a low-order discontinuity at the initial point which propagates because of the effects of delays. The exception is that sometimes there are low-order discontinuities in the history function, too. For example, one of the components of the solution of the Marchuk immunology problem solved as Exercise 4.8 of [10] has max(0, t + 10^{-6}) as its history for t \leq 0. The discontinuity in the first derivative at t = -10^{-6} propagates into the interval of integration. The history function is discontinuous at t_0 for Example 4.4.4 of [10]. The solver can be told about points of discontinuity at points known in advance by providing them in a vector as the value of the option JUMPS. The default is that there are no points of this kind. For the sake of clarity, we describe here only the most common situation of discontinuities propagating from t_0.

For each delay the solver starts with an event function \( g = \beta_j(t, y(t)) - t_0 \). Each time a root \( t^* \) is located, an event function \( g = \beta_j(t, y(t)) - t^* \) is added. For non-neutral problems it is assumed that the solution is smoothed as the integration progresses, so this process is continued until the level of smoothing exceeds the order of the Runge–Kutta method. If the JUMPS option is used, the tracking is continued to one higher level because the solution itself might be discontinuous at such points. The Runge–Kutta method of DDE_SOLVER has an interpolant (continuous extension), meaning that it approximates a solution component not just at the end of a step, but throughout the span of a step by a polynomial. These polynomials are used in conjunction with a root-finding subroutine to locate discontinuities following each successful step. Though this means that the solver will sometimes interpolate near a discontinuity, special precautions are taken at such a point to minimize the effect of the discontinuity.

8. Illustrative computations

Chapter 4 of [10] has examples that represent a wide variety of problems for DDEs. Though most are solved with dde23, there is a section devoted to other kinds of DDEs and software that includes examples of singular and neutral DDEs solved with DKLAG6. We have solved all these examples with DDE_SOLVER. In most cases the numerical results were exported to MATLAB and plotted there. We have written M-files for this purpose. Besides showing how easy it is to solve even rather complicated tasks with DDE_SOLVER, the examples can serve as templates. All these examples and the module for DDE_SOLVER are available at the web site [5].

Models of self-pulsing lasers with delayed feedback are formulated in Carr [1,2]. Carr has been solving these DDEs in MATLAB with dde23, but some of the computations are lengthy. Although it is not as easy to code the solution of these models with DDE_SOLVER, it is not much harder. In compensation for extra work writing the program and exporting the solution for plotting, DDE_SOLVER is dramatically faster. DKLAG6, dde23, and ddesd are all based on explicit Runge–Kutta formulas. The first two are based on pairs of orders (5, 6) and (2, 3), respectively. The third estimates its error in a different way that is roughly equivalent to a (3, 4) pair. The difference in order is significant for the tolerances of this example, but it is the difference between interpreted and compiled computation that accounts for most of reduction in run time. We illustrate the use of DDE_SOLVER with an example of a self-pulsing semiconductor laser with an external cavity [1]. The task is defined in a module that we reproduce here with all blank lines removed for brevity:
MODULE define_DDEs

IMPLICIT NONE

INTEGER, PARAMETER :: NEQN=3, NLAGS=1, NEF=1

! Physical parameters
DOUBLE PRECISION :: A1, A2, a, g, alpha, omegan, eta

CONTAINS

SUBROUTINE DDES(T,Y,Z,DYDT)

DOUBLE PRECISION :: T

DOUBLE PRECISION, DIMENSION(NEQN) :: Y, DYDT

DOUBLE PRECISION, DIMENSION(NEQN, NLAGS) :: Z

INTENT(IN) :: T, Y, Z

INTENT(OUT) :: DYDT

! Local variables
DOUBLE PRECISION :: E, D, P, ylag1, ylag3

E = Y(1)
D = Y(2)
P = Y(3)

ylag1 = Z(1,1) ! Retarded field.

ylag3 = Z(3,1) ! Retarded phase.

IF (T < 200D0) THEN
    eta = 0D0
ELSIF (T < 200D0) THEN
    eta = 0.2D0
END IF

DYDT(1) = (0.5D0)*(D + A2/(1D0 + a*E**2) - 1D0)*E &
    + eta*ylag1*COS(ylag3 - P)

DYDT(2) = g*(A1 - (1D0 + Y(1)**2)*D)

DYDT(3) = omegan + (0.5D0*alpha)*(D + A2/(1D0 + a*E**2)) &
    + eta*(ylag1/E)*SIN(ylag3 - P)

RETURN

END SUBROUTINE DDES

SUBROUTINE EF(T,Y,DYDT,Z,G)

DOUBLE PRECISION :: T

DOUBLE PRECISION, DIMENSION(NEQN) :: Y, DYDT

DOUBLE PRECISION, DIMENSION(NEQN, NLAGS) :: Z

DOUBLE PRECISION, DIMENSION(NEF) :: G

INTENT(IN) :: T, Y, DYDT, Z

INTENT(OUT) :: G

G(1) = A1 - (1D0 + Y(1)**2)*Y(2)

RETURN

END SUBROUTINE EF

END MODULE define_DDEs
This module is USEed in the main program where the physical parameters are assigned values. For brevity we provide only portions of the program. The complete program, laserex.f90, and an M-file to plot the solution are available from the web site.

```
PROGRAM laserex
 .

USE define_DDEs
USE DDE_SOLVER_M
 .

INTEGER, DIMENSION(3) :: NVAR=(/ NEQN,NLAGS,NEF /)
TYPE(DDE_SOL) :: SOL
TYPE(DDE_OPTS) :: OPTS
DOUBLE PRECISION, DIMENSION(NEQN) :: HISTORY= &
   (/ 8.6986330579174D0, 0.11387116103411D0, 0D0 /)
INTEGER :: I,J ! Local variables
DOUBLE PRECISION, PARAMETER :: DELAY=20D0,T0=0D0,TF=400D0

! For plotting y(t) against y(t - delay) in LASEREX.M,
! the mesh must have the following form. Note that the
! last point of the integration, TSPAN(NOUT), may not be
! exactly TF.
DOUBLE PRECISION, PARAMETER :: REFINE=50,SPACING=DELAY/REFINE
INTEGER, PARAMETER :: NOUT = 1 + (TF - T0)/SPACING
DOUBLE PRECISION, DIMENSION(NOUT) :: &
   TSPAN = (/ (T0 + (I-1)*SPACING, I=1,NOUT) /)
 .

OPTS = DDE_SET(RE=1D-5,AE=1D-9,DIRECTION=(/ -1 /),&
   JUMPS=(/ 200D0 /))
SOL = DDE_SOLVER(NVAR,DDES,(/ DELAY /),HISTORY,&
   TSPAN,OPTIONS=OPTS,EVENT_FCN=EF)
 .

END PROGRAM laserex
```

The number of equations, NEQN, the number of delays, NLAGS, and the number of event functions, NEF, are defined in define_DDEs as PARAMETERs so that they can be used for dimensioning arrays in the program. The output SOL is declared to be of the DDE_SOL derived type. Options are set, so we similarly define a variable OPTS as being of type DDE_OPTS. The one delay function is \( t - 20 \), so we can define it by supplying the parameter DELAY in an array. The history function is constant, so we can define it by the array HISTORY. A variety of plots are to be made that involve components of \( y(t) \) and \( y(t - 20) \). Accordingly we define TSPAN as an array of NOUT equally spaced points. The parameter
REFINE specifies how many output points are to be returned in each interval of length DELAY. It is used to determine NOUT and the SPACING, which are subsequently used in the formation of TSPAN with an array constructor and implied DO loop. Scalar tolerances more stringent than the defaults are set using DDE_SET as options in a structure OPTS and passed to the solver using the keyword OPTIONS. The name of the event function is passed to the solver using the keyword EVENT_FCN. This event function is used to locate the local maxima of a quantity called inversion by finding where its first derivative vanishes. Minima are excluded by setting the DIRECTION option to $-1$. Notice that the parameter $\eta$ changes in the subroutine DDES from 0 to 0.2 at $t = 200$.

We do not show the portions of the program concerned with displaying information about the solution and events on the screen and writing them to files. The results were entirely consistent with those computed in MATLAB using dde23. We remark that there were 13 events.

9. Conclusions

We developed DDE_SOLVER from DKLAG6. We have applied our experience writing DDE solvers in MATLAB and the capabilities of Fortran 90 to the development of a friendly Fortran DDE solver. How we accomplished this may be of interest to others who would like to modernize legacy code.

Although accomplishing everything we wanted to do resulted in significant restructuring of the legacy code, it was worth the effort because we believe that DDE_SOLVER is superior by far to DKLAG6. A user has to provide much less information and what must be provided is more natural and can be coded in an easier way. In the course of developing a completely new user interface, we have added significantly to the capabilities of DKLAG6, capabilities that make DDE_SOLVER more comparable to dde23/sd.

dde23 is capable of solving most constant lag problems of interest. Similarly, ddesd is capable of solving most time- and state-dependent delay problems. Although solving a given problem with DDE_SOLVER is not as easy as solving it with dde23/sd, it is not much harder and the difference is largely due to writing programs in F90 instead of MATLAB. In some circumstances DDE_SOLVER is clearly the best of these three DDE solvers. It is much faster than dde23/sd, so it should be used whenever run time is important. Finally, it has been applied successfully to some singular DDEs and to some neutral DDEs; dde23/sd cannot even be applied to such problems.

Acknowledgements

We are grateful to T.W. Carr for providing the physical example of Section 8 and a MATLAB program for solving it with dde23. Our conversations about the effective solution of DDEs arising in [1,2] influenced the design of DDE_SOLVER.

References


